Das Riemannintegral

Aufgaben: 5

```
> restart;
  with(plots):
```

Riemannsche Unter- und Obersummen

MATH: Eine **Zerlegung** eines abgeschlossenen Intervalles $[a, b]$ mit reellen Zahlen $a < b$ ist eine endliche Folge

$$z = (z_i \mid i = 0,.., n) \quad \text{mit} \quad a = z_0 < z_1 < ... < z_n = b.$$

Die Zerlegung heißt **äquidistant**, falls

$$z_{i+1} - z_i = \frac{(b-a)}{n} \quad \text{ist für alle} \quad i = 0,..., n-1.$$

MATH: Für eine beschränkte reelle Funktion $f: [a, b] \to \mathbb{R}$, eine Zerlegung

$$Z: a = z_0 < z_1 < ... < z_n = b \quad \text{von } [a, b]$$

und eine Folge ξ von Zwischenpunkten $\xi_i \in [z_i, z_{i+1}]$ definiert man eine **Riemannsche Summe**

$$S(f, Z, \xi) := \sum_{i=0}^{n-1} f(\xi_i)\,(z_{i+1} - z_i),$$

welche zwischen der **Untersumme**

$$U(f, Z) := \sum_{i=0}^{n-1} m_i \cdot (z_{i+1} - z_i)$$

und der **Obersumme**

$$\mathrm{O}(f, Z) := \sum_{i=0}^{n-1} M_i \cdot (z_{i+1} - z_i)$$

liegt, wobei $m_i := \inf(f(x) \mid x \in [z_i, z_{i+1}])$ und $M_i := \sup(f(x) \mid x \in [z_i, z_{i+1}])$ ist.

MAPLE: Hier ist eine Funktion, die die Definition der Unter- und Obersummen (gelb-rot bzw. weiß) für monoton steigende Funktionen und äquidistante Partitionen visualisiert:

```
> Rie:=proc(f,a,b,n::posint)
  local z,L1,L2,i;
    z:=map(i->a+i*(b-a)/n,[$0..n]);
    L1:=map(i->
      if f(z[i])>0 then
       polygonplot(
          [[z[i],0],[z[i+1],0],[z[i+1],f(z[i])],[z[i],f(z[i])]
  ],
          color=yellow)
      else
       polygonplot(
          [[z[i],0],[z[i+1],0],[z[i+1],f(z[i])],[z[i],f(z[i])]
  ],
```

```
        color=red)
      end if
      ,[$1..n]);
   L2:=map(i->
     polygonplot(
       [[z[i],0],[z[i+1],0],[z[i+1],f(z[i+1])],[z[i],f(z[i+1]
)]],
       color=white)
      ,[$1..n]);
   display(L1,L2,plot(f(x),x=a..b,color=black))
 end proc:
> Rie(unapply(x^2-3,x),1,3,20);
```



DENKANSTOSS: Betrachte den Fall positiver monoton steigender Funktionen und den negativer monoton steigender Funktionen getrennt an Hand einiger Beispiele.

ÜBUNG [01]:

1) Modifiziere das Programm **Rie**, so daß es für monoton fallende Funktionen

funktioniert.
2) Warum lassen sich Ober- und Untersummen für nicht monotone Funktionen nicht so einfach visualisieren?

Verfeinerungen der Zerlegung

MATH: Eine Verfeinerung einer Zerlegung $z = (z_i \mid i = 0,.., n)$ ist eine Zerlegung $z' = (z_i' \mid i = 0,.., k)$, welche die Zerlegung z als Teilfolge hat. Damit folgt sofort: $k \geq n$.

Hier ist noch eine Demonstration, die die Namenswahl erklärt.

```
> aa:=array(1..1,1..2):
> aa[1,1]:=display(Rie(unapply(x^2-3,x),1,3,15),Rie(unapply
  (x^2-3,x),1,3,5)) :
> aa[1,2]:=display(Rie(unapply(x^2-3,x),1,3,5)):
> display(aa);
```





ÜBUNG [02]:

1) Man gebe (inspiriert durch die Diagramme) den genauen Wert der Differenz zwischen Ober-und Untersumme bei monoton steigenden Funktionen an, wenn eine äquidistante Zerlegung vorliegt.
2) Zeige: Bei monoton steigenden Funktionen geht die Differenz zwischen Ober- und Untersumme gegen Null, wenn die maximale Intervallänge der Partitionen gegen Null geht und eine äquidistante Zerlegung vorliegt.

Das Riemannintegral



MATH: Falls das Supremum der Untersummen gleich dem Infimum der Obersummen ist, jeweils über **alle** Partitionen des Intervalles $[a, b]$ genommen, so heißt die Funktion f **Riemann-integrierbar** auf $[a, b]$. Der gemeinsame Wert heißt das **(bestimmte) Integral** von f über $[a, b]$:

$$\int_a^b f(x)\,dx$$

Monotone Funktionen sind Riemann-integrierbar (siehe Übung [02] **DENKANSTOSS**: Warum? Das wurde nur für äquidistante Zerlegungen gezeigt.). Stetige Funktionen sind Riemann-integrierbar.

Bei Riemann-integrierbaren Funktionen kann man das Integral über irgendeine Folge von Partitionen ausrechnen, solange die maximale Intervallänge gegen Null konvergiert.

Wir fangen aber mit einem einfacheren Beispiel an:

BEISPIEL:

```
> int(x^2,x=0..3);
```

$$9 \tag{1.3.1}$$

```
> limit(1/n*sum((i/n)^2,i=1..3*n),n=infinity);#Obersumme
  limit(1/n*sum((i/n)^2,i=0..3*n-1),n=infinity);#Untersumme
```

$$9$$

$$9 \tag{1.3.2}$$

ÜBUNG [03]:

Man bestimme *auf Grund der bisherigen Kenntnisse in den Worksheets*

$$\int_1^2 x^3\,dx\ .$$

Hinweis: Die folgende Formel darf benutzt werden:

```
> sum(i^3,i=0..n);
```

$$\frac{1}{4}\,(n+1)^4 - \frac{1}{2}\,(n+1)^3 + \frac{1}{4}\,(n+1)^2 \tag{1.3.3}$$

MATH: Zum Abschluss dieser Einführung noch eine Funktion, die zwar beschränkt, aber nicht Riemann-integrierbar ist:

$f: [0, 1] \to \mathbb{R}$ mit $f(x) = 1$ für rationales x und $f(x) = 0$ für irrationales x.

Da sowohl die rationalen Zahlen als auch die irrationalen Zahlen dicht in = liegen, sind die Untersummen allesamt gleich 0 und die Obersummen allesamt gleich 1.

DENKANSTOSS: Unterscheiden sich zwei beschränkte Funktionen auf $[a, b]$ nur durch ihre Werte an endlich vielen Stellen, so ist die eine genau dann integrierbar, wenn die andere integrierbar ist. In diesem Fall sind die Integrale gleich.

Stammfunktionen

MATH: Sei

$$f: I \to \mathbb{R}$$

eine Funktion auf dem offenen Intervall U. Eine Funktion

$$F: I \to \mathbb{R}$$

heißt **Stammfunktion** von *f*, falls *F* differenzierbar mit Ableitung *f* ist. Mit anderen Worten, die Stammfunktionen *F* von *f* sind die Lösungen der Differentialgleichung

$$F' = f \,.$$

Eine Stammfunktion muss aber nicht in jedem Fall existieren.

MATH: Hat *f* eine Stammfunktion *F*, so sind alle Stammfunktionen von *f* gegeben durch

$$F_c : x \to F(x) + c$$

mit $c \in \mathbb{R}$ beliebig. (Man beachte, der Definitionsbereich war ein Intervall. Wie würde die Antwort heißen, wenn der Definitionsbereich die disjunkte Vereinigung zweier Intervalle wäre?) Man nennt die Gesamtheit der Stammfunktionen von *f* auch das **unbestimmte Integral** von *f*.

ÜBUNG [04]:

1) Sei $f: \mathbb{R} \to \mathbb{R} : x \mapsto \sum_{i=0}^{n} a_i \cdot x^i$ für $a_i \in \mathbb{R}$ eine Polynomfunktion. Bestimme alle Stammfunktionen von *f*.

2) Zeige, dass

```
> f:=x->piecewise(x<0,0,x>=0,1);
```

$$f := x \to piecewise(x < 0, 0, 0 \le x, 1) \tag{1.4.1}$$

keine Stammfunktion auf $\mathbb{R}$ hat.

Der Fundamentalsatz der Differential- und Integralrechnung

Der von Newton und Leibniz gefundene Fundamentalsatz der Infinitesimalrechnung hat die hohe Kunst der Flächeninhaltsberechnungen des 16. Jahrhunderts und davor, bei der nur die besten etwas ausrichten konnten, zu einer Fingerübung für Gymnasiasten degradiert. Inzwischen kann man auch einfach MAPLE befragen. Um ein tieferes Verständnis kommt man aber auch mit MAPLE nicht herum. Der zentrale Trick besteht darin, nicht nur wie bislang das Integral über das feste Intervall $[a, b]$ zu nehmen, sondern über alle Intervalle der Form $[a, x]$, die in $[a, b]$ enthalten sind.

MATH: Betrachte folgende Mengen von Funktionen (alle sind Vektorräume über $\mathbb{R}$)

$$\mathrm{C}([a, b]) = \mathrm{C}^0([a, b]) := \{\, f: [a, b] \to \mathbb{R} \mid f \text{ stetig } \}$$
$$\mathrm{C}^1([a, b]) := \{\, f: [a, b] \to \mathbb{R} \mid f \text{ stetig differenzierbar } \}$$

$$\mathrm{I}([a, b]) := \{\, f: [a, b] \to \mathbb{R} \mid f \text{ integrierbar } \}.$$

Mit den Begriffen, die wir in der linearen Algebra lernen werden, ist klar, dass es sich um Teilräume des Raumes $\mathbb{R}^{[a, b]}$ handelt, wobei $a, b \in \mathbb{R}$ mit $a < b$.

MATH: Sei $f \in \mathrm{I}([a, b])$, d. h. f sei integrierbare Funktionen auf $[a,b]$. Dann gilt

$f \in \mathrm{I}([a, x])$ für alle $x \in [a, b]$ und

$$\mathrm{F}: [a, b] \to \mathbb{R} : x \to \int_a^x f(t)\, \mathrm{d}t.$$

ist wohldefiniert und heißt **Integralfunktion** von f.

BEISPIEL:

```
> f:= x-> signum(x);
```

$$f := x \to \mathrm{signum}(x) \qquad \textbf{(1.5.1)}$$

```
> F:=x->int(f(u),u=-1..x);
```

$$F := x \to \int_{-1}^{x} f(u)\, \mathrm{d}u \qquad \textbf{(1.5.2)}$$

```
> simplify(F(x));
```

$$\begin{cases} -1 - x & x \le 0 \\ -1 + x & 0 < x \end{cases} \qquad \textbf{(1.5.3)}$$

```
> plot({signum(x),F(x)},x=-3..3,y=-2..2,discont=true);
```



DENKANSTOSS: Was passiert, wenn wir einen anderen Punkt a' für die Definition von F wählen?

MATH: (**Fundamentalsatz, 1. Teil**):
1) Für $f \in \mathrm{I}([a, b])$ ist die Integralfunktion *F* stetig auf $[a, b]$, also $F \in \mathrm{C}([a, b])$:

$$\int : \quad \mathrm{I}([a, b]) \to \mathrm{C}([a, b]) : f \mapsto F := \left(x \mapsto \int_a^x f(t)\, \mathrm{d}t \right)$$

ist eine wohldefinierte (lineare) Abbildung.
2) Für $f \in \mathrm{C}([a, b])$ ist die Integralfunktion *F* differenzierbar mit Ableitung *F'* = *f:*

$$\int : \quad \mathrm{C}([a, b]) \to \mathrm{C}^1([a, b]) : f \mapsto F := \left(x \mapsto \int_a^x f(t)\, \mathrm{d}t \right)$$

ist eine wohldefinierte (lineare) Abbildung mit der Ableitung

$$' : \mathrm{C}^1([a, b]) \to \mathrm{C}([a, b]) : F \mapsto F'$$

als (linearer) Linksinverser.

Kommentar: Beide Teile sagen: Integrieren macht Funktionen **glatter**: 1) integrierbare werden stetig; 2) stetige werden 1 Mal stetig differenzierbar.
Umformulierung von 2):
2'):

$$\frac{d}{dx}\left(\int_a^x f(u)\, \mathrm{d}u \right) = f(x)$$

für alle $x \in [a, b]$.

Beispiel:

```
> f:=x->sqrt(1-x^2):f(x);
```

$$\sqrt{-x^2+1} \qquad (1.5.4)$$

```
> Diff(Int(f(u),u=0..x),x)=simplify(diff(int(f(u),u=0..x),x));
```

$$\frac{\partial}{\partial x}\left(\int_0^x \sqrt{-u^2+1}\,\mathrm{d}u\right)=\sqrt{-x^2+1} \qquad (1.5.5)$$

MAPLE hat dies sogar als allgemeine Weisheit eingebaut:

```
> diff(Int(g(u),u=0..x),x);
```

$$g(x) \qquad (1.5.6)$$

MATH: Da Differenzieren $\mathrm{C}^1([a, b]) \to \mathrm{C}([a, b])$ linksinvers zum Integrieren $\mathrm{C}([a, b]) \to \mathrm{C}^1([a, b])$ ist, sagen unsere Prinzipien aus dem Mengen-Worksheet:

Differenzieren $\mathrm{C}^1([a, b]) \to \mathrm{C}([a, b])$ ist surjektiv

und

Integrieren $\mathrm{C}([a, b]) \to \mathrm{C}^1([a, b])$ ist injektiv.

Mit anderen Worten: Jede stetige Funktion auf $[a, b]$ ist Ableitung und verschiedene stetige Funktionen haben auch verschiedene Integralfunktionen.

Aus dem Abschnitt "Stammfunktionen" wissen wir noch:

MATH: Ist $f \in \mathrm{C}([a, b])$ eine stetige Funktion auf $[a, b]$, so heißt jede Funktion $\mathrm{G} \in \mathrm{C}^1([a, b]))$ mit Ableitung $G' = f$ **Stammfunktion** von *f*. Insbesondere ist die Integralfunktion *F* eine Stammfunktion von *f*. Mit anderen Worten, die Stammfunktionen von *f* bilden die Faser von *f* unter Differentiation $\mathrm{C}^1([a, b]) \to \mathrm{C}([a, b]))$. Alle *f*-Stammfunktionen unterscheiden sich von der Integralfunktion *F* durch eine additive Konstante. Also:

Fundamentalsatz (2. Teil): Ist F eine Stammfunktion von $f \in \mathrm{C}([a, b])$, so gilt insbesondere

$$\int_a^b f(x)\,\mathrm{d}x = F(b) - F(a).$$

Die Flächeninhaltsberechunungen, die in der Einleitung dieses Abschnittes angesprochen wurden, haben sich auf die Aufgabe reduziert, eine Stammfunktion zu finden, also eine Lösung der Differentialgleichung $F' = f$ auf $[a, b]$ zu finden.

DENKANSTOSS: Will man kurz sagen, dass Differenzieren und Integrieren invers zueinander sind, kann man $\mathrm{C}^1([a, b])$ entweder durch die Teilmenge (präziser den Teilraum)

$$\{G \in \mathrm{C}^1([a, b|)| \ G(a) = 0\}$$

ersetzen oder, mit der linearen Algebra durch den **Faktorraum**

$$\mathrm{C}^1([a, b]) \,/\, \{\text{ konstante Funktionen : } [a, b] \to \mathbb{R}\}.$$

Ersteres ist eine saubere Lösung, zweiteres eine bequeme Lösung, die auch MAPLE wählt: Die Stammfunktionen einer Funktion $f \in \mathrm{C}([a, b])$ werden als eine Funktion angesehen, die bis auf eine additive Konstante bestimmt ist. (Diese bis auf additive Konstante definierten Funktionen sind die Elemente des obigen Faktorraumes.)

MAPLE: Maple bestimmt Integrale symbolisch, indem es eine Stammfunktionen sucht. Der Maple-Befehl für die Stammfunktion ist entsprechend:

```
> f:=`f`:
> int(f(x),x);
```

$$\int f(x)\, \mathrm{d}x \qquad \textbf{(1.5.7)}$$

```
> Int(x^2,x)=int(x^2,x);
```

$$\int x^2\, \mathrm{d}x = \frac{1}{3}\, x^3 \qquad \textbf{(1.5.8)}$$

```
> dsolve(diff(G(x),x)=x^2,G(x));
```

$$G(x) = \frac{1}{3}\, x^3 + \_C1 \qquad \textbf{(1.5.9)}$$

Man sieht, Maple unterdrückt die Integrationskonstante bei der Stammfunktion und wählt eine Stammfunktion nach seinem Gusto aus.

ÜBUNG [05]:

1) Fasse zusammen, warum der Fundamentalsatz wichtig ist. Warum kann man mit Stammfunktionen eine die Fläche unter gewissen (welche?) Kurven ausrechnen?
2) Leite eine kompliziert aussehende Identität her durch Integration der Ableitung der Funktion $\sin(x)^2 \cdot \left(\exp(x^2) + 1\right) + \cos(x)^2$.
Hinweis: Der Maple-Befehl `int` ist erlaubt. Teste, ob auch wirklich eine Identität vorliegt, weil ja immer noch die Integrationskonstante unklar ist.

Integrationstechniken

Aufbauend auf den Abschnitten: "Definition des Riemannintegrals", "Der Fundamentalsatz der Differential- und Integralrechnung"

Aufgaben: 3

Obschon MAPLE gute Routinen hat, um Integrale bzw. Stammfunktionen zu bestimmen, muß man doch die beiden wesentlichen Techniken kennen, wie man dies von Hand machen kann, da diese auch in theoretischen Betrachtungen eine Rolle spielen kann. Übrigens ist die Frage, ob eine sogenannte elementare Funktion auch ein elementares Integral hat, ein algebraisches Problem, das algorithmisch entschieden werden kann (Risch-Algorithmus, vgl. Bronstein: Symbolic Integration I, Springer). In MAPLE sind die wichtigsten Teile des Algorithmus implementiert.

```
> restart;
```

Partielle Integration

MATH: Eine der wichtigsten Regeln für die Differentiation ist die Produktregel:

$$(uv)' = u'v + uv'$$

für $u, v \in C^1([a, b])$.

Mit Hilfe des Fundamentalsatzes ergibt sich hieraus sofort die **partielle Integration** (**DENKANSTOSS**: Führe dies formal aus.):

$$\int_a^x u'(t)v(t)\,dt = u(x)v(x) - u(a)v(a) - \int_a^x u(t)v'(t)\,dt$$

```
> Int(diff(u(t),t)*v(t),t=a..x)= u(x)*v(x) - u(a)*v(a)- Int(u(t)*diff(v(t),t),t=a..x);
```

$$\int_a^x \left(\frac{d}{dt} u(t)\right) v(t)\, dt = u(x)\, v(x) - u(a)\, v(a) - \left(\int_a^x u(t) \left(\frac{d}{dt} v(t)\right) dt\right) \quad \textbf{(2.1.1)}$$

für alle $x \in [a, b]$. Wenn man nur von Stammfunktionen spricht, schreibt man meistens nur

$$\int^x u'(t)v(t)\,dt = u(x)v(x) - \int^x u(t)v'(t)\,dt$$

Eine typische Anwendung der partiellen Integration ist der Abbau von Polynomfaktoren bei der Integration von Produkten von Polynomfunktionen mit anderen leicht integrierbaren Funktionen:

Beispiel:

```
> int(exp(x)*x,x);
```

$$(x-1)\, e^x \quad \textbf{(2.1.2)}$$

MATH: Wenn es mit dem Abbau von Faktoren nicht funktioniert, so kann man hoffen, daß nach ein- oder mehrmaliger Anwendung der partiellen Integration ein Vergleich möglich ist.

ÜBUNG [06]:

1) Integriere $x \to x^2 \cdot \sin(x)$, ohne den **int**-Befehl auf eine Produktfunktion anzuwenden.

2) Integriere die Funktion $\sin(x) \cdot \exp(x)$ (nach dem obigen Hinweis), ohne den **int**-Befehl auf eine Produktfunktion anzuwenden. Interpretiere den Faktor 1/2.

(Hinweis: Da man mehrfach partiell integrieren muß, lohnt es sich schon die partielle Integrationsformel als Programm zu schreiben.)

Integration durch Substitution

MATH: Eine weitere sehr wichtige Differentiationsregel ist die Kettenregel:

$\frac{du(v(x))}{dx} = u'(v(x)) \cdot v'(x)$ oder $(u \circ v)' = (u' \circ v) \cdot v'$

```
> diff(u(v(x)),x);
```

$$D(u)(v(x)) \left(\frac{d}{dx} v(x)\right) \tag{2.2.1}$$

Dieser steht gemäß dem Fundamentalsatz die Substitutionsregel beim Integrieren gegenüber:

$$\int^{x} u(v(t))\, v'(t)\, dt = \int^{v(x)} u(y)\, dy$$

oder

$$\int_{a}^{x} u'(v(t))\, v'(t)\, dt = u(v(x)) - u(v(a))$$

(**DENKANSTOSS**: Führe dies formal aus.)

```
> Int(diff(u(v(t)),t),t=a..x)=u(v(x))-u(v(a));
```

$$\int_{a}^{x} D(u)(v(t)) \left(\frac{d}{dt} v(t)\right) dt = u(v(x)) - u(v(a)) \tag{2.2.2}$$

Einfache Beispiele:

```
> int(sin(a*x),x);
```

$$-\frac{\cos(a\,x)}{a} \tag{2.2.3}$$

```
> int(sin(x^2)*x,x);
```

$$-\frac{1}{2}\cos(x^2) \tag{2.2.4}$$

```
> int(sin(cos(x))*sin(x),x);
```

$$\cos(\cos(x)) \tag{2.2.5}$$

ÜBUNG [07]:

Integriere $x \rightarrow \frac{x \cdot \cos(x^2)}{\sin(x^2)^{10}}$ unter Benutzung der Substitutionsregel. (Vorsicht bei Nullstellen im Nenner!)

Integration der Umkehrfunktion

Wir brauchen ein Hilfsprogramm:

```
> #Funktioniert nur mit neueren Maple-Version
  FillBetweenCurves:=proc(f1, f2, {[color, colour] := red, transparency := 0.0})
> local p, c, n, pts1, pts2, polys, i, x, ytop, ybottom, yvals;
>   p:=plot([f1, f2], _rest, _options['color'],numpoints=51, adaptive=false);
>   c:=select(type, p, 'specfunc'('anything', 'CURVES'));
>   pts1:=op([1,1], c);pts2:=op([2,1], c);
    n:=LinearAlgebra[RowDimension](pts1);
>   polys:=NULL;x:=Vector(n, datatype=float);ytop:=Vector(n, datatype=float);ybottom:=Vector(n, datatype=float);
>   for i to n do
>     x[i]:=pts1[i,1];
>     yvals:=pts1[i,2],pts2[i,2];
>     ytop[i]:=max(yvals);
>     ybottom[i]:=min(yvals);
>   end do;
>   polys := [seq([x[i], ytop[i]], i=1..n), seq([x[i], ybottom[i]], i=n..1, -1),[x[1], ytop[1]]];
>   plots[display](plottools[polygon](polys, _options['color'],_options['transparency'],filled=true), p);
> end proc:
```

Wir haben bereits die Ableitung einer Umkehrfunktion untersucht. Nun soll es um Integration der Umkehrfunktion gehen.

MATH: (Integration der Umkehrfunktion) Sei $f\colon [a, b] \rightarrow \mathbb{R}$ stetig und bijektiv und g die Umkehrfunktion von f. Dann gilt:

$$\int_{f(a)}^{f(b)} g(y)\,\mathrm{d}y = b \cdot f(b) - a \cdot f(a) - \int_a^b f(x)\,\mathrm{d}x$$

Wir deuten die Idee der Formel an einem Beispiel an. Sei dazu $f\colon [a, b] \rightarrow \mathbb{R}_{>0} : x \mapsto x \cdot e^x$ mit $a = \frac{1}{2}$ und $b = 1$. Man sieht leicht ein, dass f injektiv (und damit bijektiv auf sein Bild) ist, indem man die Ableitung von f bildet.

```
> f:=x->exp(x)*x;
  a:=1/2:
  b:=1:
```

(2.3.1)

$$f := x \rightarrow e^{x} x \qquad \textbf{(2.3.1)}$$

Wir kennen bereits die rote Fläche unter dem Graphen von f in dem Intervall $[a, b]$, da wir durch partielle Integration die Stammfunktion von f gefunden haben:

```
> plots[display](FillBetweenCurves(0,f(x),x=a..b,color=red,
  transparency=0.25,scaling=constrained,color=red,thickness=2,
  filled=true),plot(f,0..b,scaling=constrained,color=black,
  thickness=2));
```



Wir können aber noch nicht die Stammfunktion der Umkehrfunktion g von f berechnen :

```
> g:=unapply(solve(f(x)=y,x),y);
```

$$g := y \rightarrow \mathrm{LambertW}(y) \qquad \textbf{(2.3.2)}$$

Wir nutzen jetzt einen Trick, um den Inhalt

$$\int_{f(a)}^{f(b)} g(y)\,\mathrm{d}y$$

der blauen Flächen im folgenden Plot zu bestimmen.

```
> plots[display](FillBetweenCurves(max(f(a),f(x)),f(b),x=0..b,
  color=blue,transparency=0.5,scaling=constrained),
```

```
FillBetweenCurves(0,f(x),x=a..b,color=red,transparency=0.5,
scaling=constrained),plot(f,0..b,color=black,thickness=2));
```



Die Vereinigung der roten und der blauen Fläche ist gleich

$$\int_a^b f(x)\,\mathrm{d}x + \int_{f(a)}^{f(b)} g(y)\,\mathrm{d}y = b \cdot f(b) - a \cdot f(a),$$

wobei $a \cdot f(a)$ der Flächeninhalt des weißen Rechteckes links unten ist und $b \cdot f(b)$ der Flächeninhalt des gesamten Plots.

Hier können wir das erste Integral und die komplette rechte Seite ausrechnen.

DENKANSTOSS: Die Formel läßt sich auch direkt aus der Produktregel und Substitutionsregel herleiten. Idee: $\int x\,\mathrm{d}y = x \cdot y - \int y\,\mathrm{d}x$

ÜBUNG [08]:

Bestimme $\int_0^1 \arcsin(x)\,\mathrm{d}x$ ohne den **int**-Befehl zu verwenden.